QUARTIC – Solving Quartic Equations

Abstract

QUARTIC is a program written in 1979 for the HP-34C programmable calculator to compute non-iteratively all four real and/or complex roots of a general quartic (4th-degree) polynomial equation with real coefficients. Will also work as-is in the HP-67/97 and in the HP-15C, the HP-41C series and other RPN models with minimal changes. Four worked examples are included.

Keywords: *quartic, polynomial equation, non-iterative, programmable calculator, RPN, HP-34C, HP-67, HP-97, HP-41C, HP-15C*

1. Introduction

QUARTIC is a *175-step RPN* program that I wrote in 1979 for the *HP-34C* calculator (will also run *as-is* in the *HP-67/97* and with minor changes *(see **Note 4**)* in other RPN models such as the *HP-15C* and *HP-41C*), which finds at once all real and/or complex roots of a polynomial equation of degree 4 or below, using exact formulas.

The **SOLVE** keyword is <u>not</u> used. The advantages of using this program instead are three-fold:

1. the algorithm is non-iterative and *global*, i.e.: it doesn't require initial guesses for the roots,
2. it takes only 15-20 seconds to compute all roots with full precision regardless of the display setting, and
3. it computes all roots at once, whether real or complex.

The equation to solve is a ***quartic*** of the general form:

$$x^4 + b\,x^3 + c\,x^2 + d\,x + e = 0$$

First of all, it's reduced to this depressed form, which lacks the x^3 term:

$x_1^4 + p\,x_1^2 + q\,x_1 + r = 0$, where we have: $k = b/4$, $p = (c - 6k^2)/2$, $q = 2k(4k^2 - c) + d$, $r = k(k(c - 3k^2) - d) + e$

Now the following *auxiliary cubic* equation in a is solved:

$$a^3 + 2p\,a^2 + (p^2 - r)\,a - q^2/8 = 0$$

Let a be a real root of this cubic, then the 4 roots of the quartic are found by solving these two *quadratic* equations:

$$x_1^2 - \sqrt{2a}\;x_1 + (p + a + q/2\sqrt{2a}) = 0$$
$$x_1^2 + \sqrt{2a}\;x_1 + (p + a - q/2\sqrt{2a}) = 0$$

We thus get four roots, real and/or complex, and the four roots of the *original quartic* are then: $x = x_1 - k$
The *auxiliary cubic* equation is solved like this:

let $B = 2p$, $C = p^2 - r$, $D = -q^2/8$, then we have: $P = C - B^2/3$, $Q = D + B(2B^2/27 - C/3)$

and the cubic gets reduced to this depressed form, which lacks the z^2 term:

$$z^3 + P\,z + Q = 0$$

One root of this cubic is computed as follows:

let $D = (Q/2)^2 + (P/3)^3$,

now, if $D \geq 0$ then $z = \sqrt[3]{-\frac{Q}{2} + \sqrt{D}} + \sqrt[3]{-\frac{Q}{2} - \sqrt{D}}$, else $z = 2\sqrt{-\frac{P}{3}}\,cos\left(\frac{1}{3}\,arc\,cos\,\frac{3\,Q}{2\,P}\sqrt{-\frac{3}{P}}\right)$

and finally, the root of the *auxiliary cubic* is: $a = z - B/3$

2. Program Listing

```
01 ♦LBL A     36  -        71  2         106  x^2      141 CHS
02 RCL 1      37  x        72  ÷         107 RCL 4     142 ♦LBL 8
03 RCL 0      38 STO+ 3    73 STO 1      108  -        143 RCL I
04  4         39 RCL 4     74 x^2        109 RCL 0     144  -
05  ÷         40  x^2      75 RCL 3      110 CHS       145 RTN
06 STO I      41 RCL 3     76  3         111 STO 0     146 ♦LBL 1
07  x^2       42  -        77 STO÷ 2     112  4        147 x<0
08  6         43 STO 1     78  y^x       113  ÷        148 SF 0
09  x         44 RCL 4     79  +         114  +        149 ABS
10  -         45  2        80 x<0        115 x<0       150  3
11  2         46  x        81 GTO 2 ►    116 GTO 4 ►   151 1/x
12  ÷         47 STO 2     82 √x         117 √x        152  y^x
13 STO 4      48  x^2      83 STO 3      118 STO 2     153 F? 0
14 RCL I      49  3        84 RCL 1      119 RCL 1     154 CHS
15  x^2       50  ÷        85  -         120 CHS       155 CF 0
16  4         51  -        86 GSB 1 ►    121 STO 1     156 RTN
17  x         52 STO 3     87 RCL 1      122 CHS       157 ♦LBL 2
18 RCL 1      53 RCL 2     88 CHS        123  +        158 RCL 1
19  -         54  x^2      89 RCL 3      124 GSB 8 ►   159 RCL 3
20 RCL I      55  2        90  -         125 R/S       160 1/x
21  x         56  x        91 GSB 1 ►    126 RCL 1     161 CHS
22  2         57  9        92  +         127 CHS       162 √x
23  x         58  ÷        93 ♦LBL 3     128 RCL 2     163 STO 1
24 RCL 2      59 RCL 1     94 RCL 2      129  -        164  x
25  +         60  -        95  -         130 GTO 8 ►   165 RCL 3
26 STO 0      61 RCL 2     96 STO+ 4     131 ♦LBL 4    166  ÷
27 RCL I      62  x        97  2         132 CHS       167 COS^-1
28 RCL 1      63  3        98  ÷         133 √x        168  3
29 RCL I      64 STO÷ 3    99 √x         134  1        169  ÷
30  x^2       65  ÷        100 STO 1     135 PSE       170 COS
31  3         66 RCL 0     101 STO÷ 0    136 x↔y       171 RCL 1
32  x         67  x^2      102 GSB 6 ►   137 R/S       172  ÷
33  -         68  8        103 R/S       138 RCL 1     173  2
34  x         69  ÷        104 ♦LBL 6    139 CHS       174  x
35 RCL 2      70  -        105 RCL 1     140 STO 1     175 GTO 3 ►
```

- *175 program steps*
- *uses registers R_0 - R_4 , R_I*
- *uses labels A,1,2,3,4,6,8*
- *uses flag 0*
- *angular mode is irrelevant*
- *any display setting*

- *the symbols ♦ and ► are purely cosmetic to indicate branching*

3. Usage Instructions

Step 1: Store the coefficients: *b* **STO 0** *c* **STO 1** *d* **STO 2** *e* **STO 3**

Step 2: To find the four roots, press **A** → *{if the roots are real nothing pauses and they are shown next}*

→ x_1 **R/S** → x_2 **R/S** → x_3 **R/S** → x_4

If any of the two pairs of roots (or both) are *complex* then an indicator "**1**" will be paused for one second, immediately followed by the imaginary and real parts, like this:

press **A** → *{if the roots are complex, the indicator "**1**" is paused, then their components}*

→ **1** → *imaginary part* **R/S** → *real part*

and the two conjugate roots are: $x_{1,2}$ = *real part* ± *imag. part* . ***i***

To solve another equation go to *Step 1* above.

4. Examples

The following examples can be useful to check that the program is correctly entered and to understand its usage.

4.1 Example 1

Find all roots of the quartic equation: $x^4 - 10x^3 + 35x^2 - 50x + 24 = 0$

Store the coefficients: -10 `STO 0` 35 `STO 1` -50 `STO 2` 24 `STO 3`

Compute the roots: `A` → *(no **1** indicator pauses ever so all four roots are real)*

*(assume **FIX 4** display setting)* → *4.0000* `R/S` → *3.0000* `R/S` → *2.0000* `R/S` → *1.0000*

so the four (real) roots are: $x_1 = 4.0000$, $x_2 = 3.0000$, $x_3 = 2.0000$, $x_4 = 1.0000$

4.2 Example 2

Find all roots of the quartic equation: $x^4 + 3x^3 + 8x^2 + 7x + 5 = 0$

Store the coefficients: 3 `STO 0` 8 `STO 1` 7 `STO 2` 5 `STO 3`

Compute the roots: `A` → ***1.0000*** *(the **1** indicator pauses, so the **first** 2 roots are complex)*

*(assume **FIX 4** display setting)* → *0.8660* *(the imaginary part)*

`R/S` → *-0.5000* *(the real part)*

`R/S` → ***1.0000*** *(the indicator pauses again so the **2nd** pair are complex too)*

→ *2.0000* *(the imaginary part)*

`R/S` → *-1.0000* *(the real part)*

so the four (complex) roots are: $x_{1,2} = -0.5000 \pm 0.8660\,\boldsymbol{i}$, $x_{3,4} = -1.0000 \pm 2.0000\,\boldsymbol{i}$

4.3 Example 3

Find all roots of the quartic equation: $x^4 - 2x^3 - x + 2 = 0$ *(notice there's no x^2 the term)*

Store the coefficients: -2 `STO 0` 0 `STO 1` -1 `STO 2` 2 `STO 3`

Compute the roots: `A` → *2.0000* *(no **1** indicator pauses, so this is the first real rootl)*

*(assume **FIX 4** display setting)* `R/S` → *1.0000* *(the second real root)*

`R/S` → ***1.0000*** *(the **1** indicator pauses so the final 2 roots are complex)*

→ *0.8660* *(the imaginary part)*

`R/S` → *-0.5000* *(the real part)*

so the four roots are: $x_1 = 2.0000$, $x_2 = 1.0000$, $x_{3,4} = -0.5000 \pm 0.8660\,\boldsymbol{i}$

4.4 Example 4

Find all roots of the ***cubic*** equation: $x^3 - 6x - 2 = 0$ *(multiplying by **x** we get the quartic* $x^4 - 6x^2 - 2x = 0$*)*

Store the coefficients: 0 **STO 0** -6 **STO 1** -2 **STO 2** 0 **STO 3**

Compute the roots: **A** → *(no **1** indicator pauses ever so all roots are real)*

*(assume **FIX 9** display setting)* → *2.601679132* *(the first real root)*

R/S → *0.000000000* *(spurious **0** root due to the conversion to a quartic)*

R/S → *-0.339876887* *(the second real root)*

R/S → *2.261802245* *(the third real root)*

so, ignoring the spurious one, the roots are: $x_1 = 2.601679132$, $x_2 = -0.339876887$, $x_3 = -2.261802245$

Notes

*1.*If the equation to solve is a *quartic* but the x^4 term has a coefficient distinct of ***1***, simply divide all coefficients by it.

*2.*It can also be used to solve *quadratic* and *cubic* equations, simply multiply the equation times ***x*** or x^2 to turn it into a quartic, find all four roots and discard the spurious ones that are ***0*** or very close to ***0***. See *Example 4.4* above.

*3.*If the equation has multiple roots the procedure may fail or the accuracy may be affected.

*4.*This program will run *as-is* on the *HP-67/97*, no changes required. It will also run in the *HP-41C* series by just changing all R_I references to R_{05} and all **GSB** instructions to **XEQ**. In the *HP-15C* it can be optimized further by using **RCL** arithmetic.

*5.*This *175 step* program effectively demonstrates the benefits of the *HP-34C'*s *expandable* program memory. For instance, though the *HP-29C* has more RAM, it can't be allocated flexibly and it's limited to 98 steps, so this program won't fit in.

*6.*This program is published in *Hewlett-Packard's Solution Book "HP-34C Matemática Avanzada" (Spanish)*. The program listing there includes keycodes for all steps and fully commented source code.

*7.*It's also published in *PPC Technical Notes V1N2 pp15-18, September 1980* (a publication of the *PPC Club Melbourne Chapter*), as *"HP-34C Polynomial Solutions"*. The program listing there includes keycodes for all steps but no comments.

References

Murray Spiegel *et al.*(1970). *Manual de Fórmulas y Tablas Matemáticas. (Schaum).*
Valentin Albillo (1980). *Hewlett-Packard's Solution Book "HP-34C Matemática Avanzada"*
Valentin Albillo (1980). *PPC Technical Notes V1N2 pp15-18*, as *"HP-34C Polynomial Solutions"*

Copyrights